wntestq — Portmanteau (Q) test for white noise

Syntax

```
wntestq varname [if] [in] [, lags(#)]
```

You must `tsset` your data before using `wntestq`; see [TS] **tsset**. Also the time series must be dense (nonmissing with no gaps in the time variable) in the specified sample.

varname may contain time-series operators; see [U] **11.4.4 Time-series varlists**.

Menu

Statistics > Time series > Tests > Portmanteau white-noise test

Description

`wntestq` performs the portmanteau (or Q) test for white noise.

Option

`lags(#)` specifies the number of autocorrelations to calculate. The default is to use $\min(\lfloor n/2 \rfloor - 2, 40)$, where $\lfloor n/2 \rfloor$ is the greatest integer less than or equal to $n/2$.

Remarks and examples

stata.com

Box and Pierce (1970) developed a portmanteau test of white noise that was refined by Ljung and Box (1978). See also Diggle (1990, sec. 2.5).

▷ Example 1

In the example shown in [TS] **wntestb**, we generated two time series. One (`x1`) was a white-noise process, and the other (`x2`) was a white-noise process with an embedded cosine curve. Here we compare the output of the two tests.

```
. drop _all
. set seed 12393
. set obs 100
obs was 0, now 100
. generate x1 = rnormal()
. generate x2 = rnormal() + cos(2*_pi*(_n-1)/10)
. generate time = _n
. tsset time
        time variable:  time, 1 to 100
                delta:  1 unit
```

```
. wntestb x1, table
Cumulative periodogram white-noise test
---------------------------------------
 Bartlett's (B) statistic  =     0.7093
 Prob > B                  =     0.6957
. wntestq x1
Portmanteau test for white noise
---------------------------------------
 Portmanteau (Q) statistic =    32.6863
 Prob > chi2(40)           =     0.7875
. wntestb x2, table
Cumulative periodogram white-noise test
---------------------------------------
 Bartlett's (B) statistic  =     1.8323
 Prob > B                  =     0.0024
. wntestq x2
Portmanteau test for white noise
---------------------------------------
 Portmanteau (Q) statistic =   129.4436
 Prob > chi2(40)           =     0.0000
```

This example shows that both tests agree. For the first process, the Bartlett and portmanteau tests result in nonsignificant test statistics: a p-value of 0.9053 for `wntestb` and one of 0.9407 for `wntestq`.

For the second process, each test has a significant result to 0.0010.

◁

Stored results

`wntestq` stores the following in `r()`:

Scalars

`r(stat)`	Q statistic	`r(p)`	probability value
`r(df)`	degrees of freedom		

Methods and formulas

The portmanteau test relies on the fact that if $x(1), \ldots, x(n)$ is a realization from a white-noise process. Then

$$Q = n(n+2) \sum_{j=1}^{m} \frac{1}{n-j} \, \widehat{\rho}^{\,2}(j) \longrightarrow \chi^2_m$$

where m is the number of autocorrelations calculated (equal to the number of lags specified) and $\longrightarrow$ indicates convergence in distribution to a χ^2 distribution with m degrees of freedom. $\widehat{\rho}_j$ is the estimated autocorrelation for lag j; see [TS] **corrgram** for details.

References

Box, G. E. P., and D. A. Pierce. 1970. Distribution of residual autocorrelations in autoregressive-integrated moving average time series models. *Journal of the American Statistical Association* 65: 1509–1526.

Diggle, P. J. 1990. *Time Series: A Biostatistical Introduction*. Oxford: Oxford University Press.

Ljung, G. M., and G. E. P. Box. 1978. On a measure of lack of fit in time series models. *Biometrika* 65: 297–303.

Sperling, R. I., and C. F. Baum. 2001. sts19: Multivariate portmanteau (Q) test for white noise. *Stata Technical Bulletin* 60: 39–41. Reprinted in *Stata Technical Bulletin Reprints*, vol. 10, pp. 373–375. College Station, TX: Stata Press.

Also see

[TS] **tsset** — Declare data to be time-series data

[TS] **corrgram** — Tabulate and graph autocorrelations

[TS] **cumsp** — Cumulative spectral distribution

[TS] **pergram** — Periodogram

[TS] **wntestb** — Bartlett's periodogram-based test for white noise